

# *Two Compositions for Computer*

by

Justin Henry Rubin

The following programs were composed in December 1994, on the Atari 800XL using BASIC. Each of the works use restrictive random operations within compositional parameters set by the author. This allows results that are audibly different with each execution of the programs while retaining identical structural principals.

For analysis purposes or if one wishes to enter these programs into another system, the following information is necessary:

1. *rnd(0)\*x* .....a random number is chosen between 0 and x
2. *int(x)* .....transforms x into its closest integer
3. *sound x,y,z,q* ...produces a sound from channel x (0-3), at pitch y (1-255), with z distortion (10 and 14 are sine waves), with dynamic q (0-15).

---

## *Staggered Music*

5 PRINT "Staggered Music by Justin Rubin"

10 FOR JQ=1 TO 2

*10-20 divides piece into two sections*

15 IF JQ= 1 THEN 5000

20 IF JQ= 2 THEN 6000

*60-400 determines in each voice separately  
if the pitch should change, and if so,  
determines what the new pitch value is*

60 E=RND(0)\*10

*60-100 voice 1 (a)*

70 IF E<7 THEN 160

100 A=RND(0)\*T

160 F=RND(0)\*10                    *160-200 voice 2(b)*

170 IF F<6 THEN 260

200 B=RND(0)\*T

260 G=RND(0)\*10                    *260-300 voice 3(c)*

270 IF G<5 THEN 360

300 C=RND(0)\*T

360 H=RND(0)\*10                    *360-400 voice 4(d)*

370 IF H<4 THEN 440

400 D=RND(0)\*T

*440-470 determines the dynamic for each  
voice*

440 I=RND(0)\*15

450 J=RND(0)\*15

460 K=RND(0)\*15

470 L=RND(0)\*15

*500-530 execution of sound*

500 SOUND 1,A,14,I

510 SOUND 2,B,14,J

520 SOUND 3,C,10,K

530 SOUND 0,D,10,L

*540-600 computes average pitch of the*

*voices*

540 X=A+B+C+D

600Z=X/4

*700-850 sets variables for staggering  
depending on the average pitch (Z)*

700 IF Z>200 THEN GOSUB 2000 *all four stagger*

750 IF Z>150 AND Z<200 THEN GOSUB 2010 *three stagger*

800 IF Z>100 AND Z<150 THEN GOSUB 2015 *two stagger*

850 IF Z>50 AND Z<100 THEN GOSUB 2020 *only one staggers*

*1000-1010 loops operations*

1000 NEXT T  
1010 NEXT JQ

*1600-1640 ends composition*

1600 SOUND 1,0,0,0  
1610 SOUND 2,0,0,0  
1620 SOUND 3,0,0,0  
1630 SOUND 0,0,0,0  
1640 END

*2000-2040 execution of staggering*

2000 PA=RND(0)\*4: PPA=INT(PA) *determines which voice is to stagger*  
2004 SOUND PPA,0,0,0 *stagger operation*  
2010 PB=RND(0)\*4: PPB=INT(PB)  
2012 IF PPB=PPA THEN 2010 *insures new staggered voice is not the same*  
2013 SOUND PPB,0,0,0  
2015 PC=RND(0)\*4: PPC=INT(PC)  
2016 IF PPC=PPB or PPC=PPA THEN 2015 *idem*  
2017 SOUND PPC,0,0,0  
2020 PD=RND(0)\*4: PPD=INT(PD)  
2025 IF PPD=PPC or PPD=PPB or PPD=PPA THEN 2020 *idem*  
2030 SOUND PPD,0,0,0  
2040 RETURN

*5000-6010 delineation of the two sections*

5000 FOR T=1 TO 255 *SECTION ONE (increases possible pitches)*  
5010 GOTO 60  
6000 FOR T=255 TO 1 STEP -1 *SECTION TWO (retrogrades variable in pitch)*  
6010 GOTO 60

*NB: Inverting line 5000 with 6000 is another valid execution for the form of the piece.*

---

—

*Arborescences: a two voice study after Xenakis*

1 PRINT "Arborescences: a two voice study after Xenakis by Justin Rubin"

5 FOR T=1 TO 255 *duration of piece; alter at will*

*6-11 pitch deviation parameters*

6 AA=RND(0)\*10 *bounderies of greatest possible pitch deviation*

7 BB=RND(0)\*10

10 A=RND(0)\*AA *actual pitch deviation*

11 B=RND(0)\*BB

*12-27 determines which direction pitch is to deviate and computes new pitch based on previous location of the voices*

12 Q=RND(0)\*20

13 QQ=RND(0)\*20

24 IF Q>10 THEN RA=RA-A

25 IF Q<10 THEN RA=RA+A

26 IF QQ>10 THEN RB=RB-B

27 IF QQ<10 THEN RB=RB+B

*30-35 centralized pitch*

30 RRA=RA+200

35 RRB=RB+200

*50-65 sets boundaries on tessitura*

50 IF RRA>255 THEN RRA=255-A

55 IF RRB>255 THEN RRB=255-B

60 IF RRA<10 THEN RRA=10+A

65 IF RRB<10 THEN RRB=10+B

*75-80 determines deviation in dynamic*

75 F=RND(0)\*20

76 FF=RND(0)\*20

77 IF F>10 THEN FA=FA-.5

78 IF F<10 THEN FA=FA+.5

79 IF FF>10 THEN FB=FB-.5

80 IF FF<10 THEN FB=FB+.5

*81-84 sets boundaries on dynamics*

81 IF FA>15 THEN FA=15

82 IF FB>15 THEN FB=15

83 IF FA<1 THEN FA=1

84 IF FB<1 THEN FB=1

*100-300 execution of sound and loop operations*

100 SOUND 0,RRA,10,FA

200 SOUND 1,RRB,10,FB

300 NEXT T

*400-500 ends composition*

400 FOR M=1 TO 500: NEXT M

*sustain of last sounds*

425 SOUND 0,0,0,0

450 SOUND 1,0,0,0

500 END